Memory leak under Arc/Orc on inline iterators with nested seq.
### Description

On the arc, orc, and atomicArc mm's: Inline iterators that do not return a lent type leak memory when iterating upon nested sequences. Closures that do not return lent types and inlines that do return lent types do not leak.

```nim
iterator myPairsInline*[T](twoDarray: seq[seq[T]]): (int, seq[T]) {.inline.} =
  for indexValuePair in twoDarray.pairs:
    yield indexValuePair

iterator myPairsClosure*[T](twoDarray: seq[seq[T]]): (int, seq[T]) {.closure.} =
  for indexValuePair in twoDarray.pairs:
    yield indexValuePair

template testTotalMem(iter: untyped): int =
  proc innerTestTotalMem(): int {.gensym.} =
    result = 0

    # do the same operation 100 times, which should have similar mem footprint
    # as doing it once.
    for iterNum in 0..100:
      result = max(result, getTotalMem()) # record current mem footprint

      # initialize nested sequence
      var my2dArray: seq[seq[int32]] = @[]

      # fill with some data...
      for i in 0'i32..10_000:
        var z = @[i, i+1]
        my2dArray.add z

      # use that data somehow...
      var otherContainer: seq[int32] = @[]
      var count = 0'i32
      for oneDindex, innerArray in my2dArray.iter:
        for value in innerArray:
          inc count
          if oneDindex > 50 and value < 200:
            otherContainer.add count

  innerTestTotalMem()

proc main =
  let closureMem = testTotalMem(myPairsClosure) #1052672
  let inlineMem = testTotalMem(myPairsInline) #20328448

  when defined(echoFootprint):
    echo "Closure memory footprint: " & $closureMem
    echo "Inline memory footprint: " & $inlineMem

  # check that mem footprint is relatively similar b/t each method
  doAssert (closureMem - inlineMem).abs < (closureMem div 10)

main()
```

### Nim Version

Nim Compiler Version 2.2.0 [Linux: amd64]

### Current Output

```text
/path/to/mainmodule.nim(49) mainmodule
/path/to/mainmodule.nim(47) main
/path/to/.choosenim/toolchains/nim-2.2.0/lib/std/assertions.nim(41) failedAssertImpl
/path/to/.choosenim/toolchains/nim-2.2.0/lib/std/assertions.nim(36) raiseAssert
/path/to/.choosenim/toolchains/nim-2.2.0/lib/system/fatal.nim(53) sysFatal
Error: unhandled exception: /path/to/mainmodule.nim(47, 3) `(closureMem - inlineMem).abs < (closureMem div 10)`  [AssertionDefect]
Error: execution of an external program failed: '/path/to/mainmodule'
```


### Expected Output

_No response_

### Known Workarounds

_No response_

### Additional Information

The regions gc/mm scheme is the only other gc that fails this example.